

## PROC DOCUMENT

(Continued)

### Entry Management Statements

**copy path-1, ..., path-m to path-n;**  
Copies *path-1, ..., path-m* to *path-n*.

**move path-1, ..., path-m to path-n;**  
Moves *path-1, ..., path-m* to *path-n*.

**make path-1, ..., path-n;**  
Creates new directories named *path-1, ..., path-n*.

**import data=data-set-name to path;**  
**import grseg=grseg-name to path;**  
Imports data set, *data-set-name*, or grseg, *grseg-name*, to *path*, respectively.

**note path "text";**  
Creates a new note at *path*.

**link path-1 to path-2;**  
Creates a symbolic link from *path-1* to *path-2*.

**rename path-1 to path-2;**  
Renames *path-1* to *path-2*.

**setlabel path "text";**  
Sets the label of *path*.

**delete path-1, ..., path-n;**  
Deletes *path-2, ..., path-n*.

**obbnote<n> path "text";**  
**obanote<n> path "text";**  
**obtitle<n> path "text";**  
**obstitle<n> path "text";**  
**obfootn<n> path "text";**  
Sets the *n*<sup>th</sup> before note, after note, title, subtitle, or footnote of *path*, respectively. If no string is specified, the note, title, or footnote is deleted.

**obpage path </<after> <delete>>;**  
Controls page breaks. The **after** option controls page breaks after *path*. The **delete** option deletes page breaks from *path*.

## Example

```
* Create a new document;
ods document name=mydocument(write);

* Run procedures;
proc contents data=sashelp.class; run;

* Close the document destination;
ods document close;

* Modify and replay the document;
proc document name=mydocument;

  * List the document to see
  * what paths exist;
  list / levels=all;
  run;

  * Move the table we want to the
  * top level, remove titles, and
  * set a new label;
  move \Contents\DataSet\Variables to ^;
  setlabel Variables "Variable Info";
  obtitle Variables;
  obstitle Variables;
  run;

  * Remove tables and directories
  * we don't need;
  delete \Contents;
  run;

  * Import the dataset as a table and
  * set a new label;
  import data=sashelp.class to Class;
  setlabel Class "Student Statistics";
  run;

  * List the document to verify changes;
  list / levels=all;
  run;

  * Open the PDF destination;
  ods pdf;

  * Replay table labeled "% Statistics";
  replay ^
    (where=( _label_ like '% Statistics' ));
  run;

  * Close the PDF destination;
  ods pdf close;

quit;
```

For complete information, refer to the Base SAS<sup>®</sup> documentation at [support.sas.com/base](https://support.sas.com/base)



# ODS DOCUMENT and PROC DOCUMENT Tip Sheet

This tip sheet places frequently used information in one place, on one sheet of paper, so you don't have to search through the online documentation. It also gives you something to take home, type in, and try.

The ODS DOCUMENT destination allows you to store a report's components in order to modify and replay them using PROC DOCUMENT. This combination of tools allows you to customize your reports far more than using ODS options and statements alone. This tip sheet presents the most common statements and options used in creating, modifying, and replaying ODS documents.



SAS Institute Inc. World Headquarters  
+1 919 677 8000 [sas.com/offices](https://sas.com/offices)

SAS and all other SAS Institute Inc. product or service names are registered trademarks or trademarks of SAS Institute Inc. in the USA and other countries. ® indicates USA registration. Other brand and product names are trademarks of their respective companies. Copyright © 2019, SAS Institute Inc. All rights reserved. 1093808.0319

# SAS<sup>®</sup>9 ODS DOCUMENT and PROC DOCUMENT Tip Sheet

## Output Objects

Output objects include tables, graphs, notes, and equations. In the case of tables, the output object consists of the data and a template name.

### ods trace on;

Displays information about output objects as they are created. This information includes the name of the object, the name of the template, labels, paths, and label paths.

### context

The context of an output object is a set of attributes that are attached to the output object. The parts of the output object context that come **before** the output object are: *before page break*, *titles*, *subtitles*, *bygroup*, and *before notes*. The parts of the output object context that come **after** the output object are: *afternotes*, *footnotes*, and *after page break*.

## Content of a Document

The content of the document includes: *output object*, *the output object context*, and *the output object hierarchy*.

The content of the document does **not** include: *proc options*, *system options*, *ODS options*, *graph options*, *GRSEGS*, and *external graph customizations*.

## Documents Window

### odsdocuments

Entering **odsdocuments** into the command bar opens the Documents window, where you can view and edit the hierarchy and replay output objects. Right-clicking and choosing **Properties** displays information about the output object such as: *type*, *name*, *description*, *template*, *the document that the output object belongs to*, *path in the hierarchy*, *file size*, *creation date and time*, and *last modified date and time*.

## Document Paths

Each output object has a path associated with it that includes the directory hierarchy, much like in a computer file system. For example:

```
\Directory-1#n\...\Directory-m#n\Object#n
```

Each segment in the path, separated by backslashes (\), consists of a name (e.g., **Directory-1**, **Directory-m**, **Object**) as well as a sequence number (e.g., **#n**). Sequence numbers are used to ensure that each directory and object is uniquely addressable.

A path that consists only of a backslash represents the root directory.

A caret (^) in a path indicates the current directory. A double caret (^ ^) indicates one directory up.

## WHERE Clauses

Document paths can be followed by a WHERE clause to further subset the objects matched by a path. The general form of WHERE clause usage is shown below.

```
document-path (where= (where-expr) )
```

Variables available in a WHERE clause are listed in the following table.

Name	Description
name	Name of the item
path	Path of the item
label	Label of the item
labelpath	Labels of all entries in path
type	Type of the item
seqno	Sequence number of the item
cdate	Creation date of the item
mdate	Modification date of the item
ctime	Creation time of the item
mtime	Modification time of the item
cdatetime	Creation date-time of the item
mdatetime	Modification date-time of the item
variable	BY variable

## Document Destination

### Creating a Document

```
ods document name=document-name;
... procedure code ...
ods document close;
```

*document-name* is the name of a document with an optional SAS<sup>®</sup> library name.

An access option can be applied to the document, *document-name*, by appending one of the following, in parentheses, to *document-name*.

#### write

Creates a new document. A document is overwritten if it already exists.

#### update

Creates a new document. A document is appended to if it already exists.

### Replaying a Document

```
proc document name=document-name;
  replay <path-1, ..., path-n> </levels=n>;
run;
```

#### replay;

Replays entire contents of document, *document-name*, to all open ODS destinations.

#### replay path-1, ..., path-n;

Replays only specified paths, *path-1*, ..., *path-n*, to all open ODS destinations.

#### replay / levels=n;

Replays all output objects in *document-name*, within *n* levels of directories, to all open ODS destinations.

#### replay path-1, ..., path-n / levels=n;

Replays only specified paths, *path-1*, ..., *path-n*, within *n* levels of directories to all open ODS destinations.

## PROC DOCUMENT

### Basic Usage

```
proc document name=document-name;
  ... statements ...
run;
```

### Document Management Statements

#### doc library=library-name;

Lists all documents in *library-name*.

#### doc name=document-name;

Opens *document-name* for update.

#### doc close;

Closes the current document.

#### delete document-name;

Deletes *document-name*.

### Navigation Statements

#### dir;

Displays the path of the current directory.

#### dir path;

Changes the current directory to *path*.

#### list </option(s)>;

#### list path-1, ..., path-n </option(s)>;

Lists the content of the current directory or *path-1*, ..., *path-n*, respectively. The following options are available.

Option	Description
bygroups	Displays BY variables
details	Displays verbose information
follow	Follows links
levels=n   all	List <i>n</i> or all directory levels, respectively
order=...	Sort output by alpha, date, or insert